
A streaming API for OS processes that mirrors Elixir's Port module. ProtonStream
keeps programs, daemons, and applications launched from Erlang and Elixir contained
and well-behaved. This lightweight library kills OS processes if the Elixir process
running them crashes and if you're running on Linux, it can use cgroups to prevent
many other shenanigans.
Some features:
	Streaming API that mirrors Elixir's Port module
	Bidirectional communication with stdin/stdout/stderr
	Set cgroup controls like thresholds on memory and CPU utilization
	Start OS processes as a different user or group
	Send SIGKILL to processes that aren't responsive to SIGTERM
	With cgroups, ensure that all children of launched processes have been killed too

TL;DR
Add proton_stream to your project's mix.exs dependency list:
def deps do
  [
    {:proton_stream, "~> 1.8"}
  ]
end
Streaming API
The primary API is ProtonStream.open/3 which starts a GenServer managing an OS
process. The calling process becomes the "owner" and receives messages:
{:ok, ps} = ProtonStream.open("cat", [])

# Send data to stdin
send(ps, {self(), {:command, "hello"}})

# Receive stdout
receive do
  {^ps, {:data, data}} -> IO.puts("Got: #{data}")
end

# Close when done
send(ps, {self(), :close})
Messages
Messages TO ProtonStream (from owner):
	{pid, {:command, binary}} - send data to child's stdin
	{pid, :close} - close the port
	{pid, {:connect, new_pid}} - transfer ownership to another process

Messages FROM ProtonStream (to owner):
	{pid, {:data, data}} - stdout from child process
	{pid, {:error, data}} - stderr from child process
	{pid, :closed} - reply to close request
	{pid, :connected} - reply to connect request
	{:EXIT, pid, reason} - process termination (when trapping exits)

Bidirectional Communication
ProtonStream supports full bidirectional communication:
{:ok, ps} = ProtonStream.open("bc", ["-q"])

send(ps, {self(), {:command, "2 + 2\n"}})
receive do
  {^ps, {:data, "4\n"}} -> :ok
end

send(ps, {self(), {:command, "10 * 10\n"}})
receive do
  {^ps, {:data, "100\n"}} -> :ok
end
Callback Module Mode
For integration with supervision trees, use ProtonStream.start_link/5 with a callback
module that implements the ProtonStream behaviour:
defmodule MyWorker do
  use GenServer
  @behaviour ProtonStream

  def start_link(args) do
    ProtonStream.start_link(__MODULE__, "my_command", [], args, [])
  end

  @impl GenServer
  def init(args) do
    {:ok, %{buffer: "", args: args}}
  end

  @impl ProtonStream
  def handle_stdout(data, state) do
    {:noreply, %{state | buffer: state.buffer <> data}}
  end

  @impl ProtonStream
  def handle_stderr(data, state) do
    IO.puts(:stderr, data)
    {:noreply, state}
  end

  @impl ProtonStream
  def handle_exit(reason, state) do
    {:stop, reason, state}
  end
end
Add to your supervision tree:
children = [
  {MyWorker, [some: :args]}
]

Supervisor.start_link(children, strategy: :one_for_one)
The callbacks handle_stdout/2, handle_stderr/2, and handle_exit/2 are all optional.
You can also implement GenServer.handle_call/3, GenServer.handle_cast/2, and GenServer.handle_info/2.
FAQ
How do I watch stdout?
Use ProtonStream.open/3 and receive {pid, {:data, data}} messages:
{:ok, ps} = ProtonStream.open("my_program", [])

# In a receive loop or GenServer handle_info:
receive do
  {^ps, {:data, data}} -> IO.write(data)
  {^ps, {:error, data}} -> IO.write(:stderr, data)
end
How do I send input to stdin?
Use the streaming API:
{:ok, ps} = ProtonStream.open("cat", [])
send(ps, {self(), {:command, "hello world\n"}})
How do I stop a ProtonStream process?
Send a close message:
send(ps, {self(), :close})
Background
The Erlang VM's port interface lets Elixir applications run external programs.
This is important since it's not practical to rewrite everything in Elixir.
Plus, if the program is long running like a daemon or a server, you use Elixir
to supervise it and restart it on crashes. The catch is that the Erlang VM
expects port processes to be well-behaved. As you'd expect, many useful programs
don't quite meet the Erlang VM's expectations.
For example, let's say that you want to monitor a network connection and decide
that ping is the right tool. Here's how you could start ping in a process.
iex> pid = spawn(fn -> System.cmd("ping", ["-i", "5", "localhost"], into: IO.stream(:stdio, :line)) end)
#PID<0.6116.0>
PING localhost (127.0.0.1): 56 data bytes
64 bytes from 127.0.0.1: icmp_seq=0 ttl=64 time=0.032 ms
64 bytes from 127.0.0.1: icmp_seq=1 ttl=64 time=0.077 ms
Now exit the Elixir process:
iex> Process.exit(pid, :oops)
true
iex> :os.cmd(~c"ps -ef | grep ping") |> IO.puts
  501 38820 38587   0  9:26PM ??         0:00.02 /sbin/ping -i 5 localhost
As you can tell, ping is still running after the exit. The reason is that
ping doesn't pay attention to stdin and doesn't notice the Erlang VM closing
it to signal that it should exit.
This is just one of the problems that proton_stream fixes.
Containment with cgroups
Even if you don't make use of any cgroup controller features, having your port
process contained can be useful just to make sure that everything is cleaned
up on exit including any subprocesses.
To set this up, first create a cgroup with appropriate permissions:
sudo cgcreate -a $(whoami) -g memory,cpu:mycgroup

Then use the cgroup options:
{:ok, ps} = ProtonStream.open("spawning_program", [],
  cgroup_controllers: ["cpu"],
  cgroup_base: "mycgroup"
)
On any error or if the Erlang VM closes the port or if spawning_program exits,
proton_stream will kill all OS processes in cgroup.
Limit memory
ProtonStream.open("memory_hog", [],
  cgroup_controllers: ["memory"],
  cgroup_base: "mycgroup",
  cgroup_sets: [{"memory", "memory.limit_in_bytes", "268435456"}]
)
Limit CPU
ProtonStream.open("cpu_hog", [],
  cgroup_controllers: ["cpu"],
  cgroup_base: "mycgroup",
  cgroup_sets: [
    {"cpu", "cpu.cfs_period_us", "100000"},
    {"cpu", "cpu.cfs_quota_us", "50000"}
  ]
)
stdio flow control
ProtonStream implements flow control to prevent the program's output from
overwhelming the Elixir process's mailbox. The :stdio_window option specifies
the maximum number of unacknowledged bytes allowed (default 10 KB).
Development
To run tests, install cgroup-tools (cgcreate, cgget):
sudo cgcreate -a $(whoami) -g memory,cpu:proton_stream_test
mix test

License
All original source code in this project is licensed under Apache-2.0.
Additionally, this project follows the REUSE recommendations
and labels so that licensing and copyright are clear at the file level.
Exceptions to Apache-2.0 licensing are:
	Configuration and data files are licensed under CC0-1.0
	Documentation is CC-BY-4.0

ProtonStream provides a streaming API for OS processes that mirrors Elixir's Port module.
This library is derived from MuonTrap by Frank Hunleth,
with a new streaming API and bidirectional stdin/stdout/stderr support.
ProtonStream can be used in two modes:
	Message-based mode - via open/3, where the calling process receives messages
	Callback module mode - via start_link/4, where a GenServer module handles events

Message-based Mode
Use open/3 to start a process. The calling process becomes the "owner" and
receives messages about stdout, stderr, and process termination.
Messages TO ProtonStream (from owner)
	{pid, {:command, binary}} - send data to child's stdin
	{pid, :close} - close the port
	{pid, {:connect, new_pid}} - transfer ownership to another process

Messages FROM ProtonStream (to owner)
	{pid, {:data, data}} - stdout from child process
	{pid, {:error, data}} - stderr from child process
	{pid, :closed} - reply to close request
	{pid, :connected} - reply to connect request
	{:EXIT, pid, reason} - process termination (when trapping exits)

Example
{:ok, ps} = ProtonStream.open("cat", [])
send(ps, {self(), {:command, "hello"}})

receive do
  {^ps, {:data, data}} -> IO.puts("Got: #{data}")
end

send(ps, {self(), :close})
Callback Module Mode
For integration with supervision trees, use start_link/4 with a callback module
that implements both GenServer and the ProtonStream behaviour.
Callbacks
The following callbacks must be implemented:
	GenServer.init/1 - called when the process starts, returns initial state
	handle_stdout/2 - called when stdout data is received
	handle_stderr/2 - called when stderr data is received
	handle_exit/2 - called when the child process exits

Example
defmodule MyWorker do
  use GenServer
  @behaviour ProtonStream

  def start_link(args) do
    ProtonStream.start_link(__MODULE__, "my_command", [], args)
  end

  @impl GenServer
  def init(args) do
    {:ok, %{buffer: "", args: args}}
  end

  @impl ProtonStream
  def handle_stdout(data, state) do
    {:noreply, %{state | buffer: state.buffer <> data}}
  end

  @impl ProtonStream
  def handle_stderr(data, state) do
    IO.puts(:stderr, data)
    {:noreply, state}
  end

  @impl ProtonStream
  def handle_exit(reason, state) do
    {:stop, reason, state}
  end
end
The module can then be added to a supervision tree:
children = [
  {MyWorker, [some: :args]}
]

Supervisor.start_link(children, strategy: :one_for_one)
Callback Return Values
	GenServer.init/1 returns {:ok, state} or {:stop, reason}
	handle_stdout/2 returns {:noreply, state} or {:stop, reason, state}
	handle_stderr/2 returns {:noreply, state} or {:stop, reason, state}
	handle_exit/2 returns {:stop, reason, state}

GenServer Callbacks
The callback module may also implement GenServer.handle_call/3,
GenServer.handle_cast/2, GenServer.handle_info/2, GenServer.handle_continue/2,
and GenServer.terminate/2 to handle synchronous requests, asynchronous requests,
messages, continuations, and cleanup.
Configuring cgroups
On most Linux distributions, use cgcreate to create a new cgroup:
sudo cgcreate -a $(whoami) -g memory,cpu:proton_stream

Then use the :cgroup_controllers and :cgroup_base options.

          @type t() :: %ProtonStream{
  args: [binary()],
  buffer: binary(),
  cgroup_path: binary() | nil,
  command: binary(),
  owner: pid(),
  port: port() | nil,
  state: term()
}

          @callback handle_exit(reason :: term(), state :: term()) ::
  {:stop, reason :: term(), new_state :: term()}

          @callback handle_stderr(data :: binary(), state :: term()) ::
  {:noreply, new_state :: term()}
  | {:noreply, new_state :: term(),
     timeout() | :hibernate | {:continue, term()}}
  | {:stop, reason :: term(), new_state :: term()}

          @callback handle_stdout(data :: binary(), state :: term()) ::
  {:noreply, new_state :: term()}
  | {:noreply, new_state :: term(),
     timeout() | :hibernate | {:continue, term()}}
  | {:stop, reason :: term(), new_state :: term()}

Returns a specification to start this module under a supervisor.
See Supervisor.

          @spec close(pid()) :: :ok


      


Close the process.
This is equivalent to send(ps, {self(), :close}).
The caller will receive {ps, :closed} when complete.

          @spec command(pid(), iodata()) :: :ok


      


Send data to the process's stdin.
This is equivalent to send(ps, {self(), {:command, data}}).

          @spec connect(pid(), pid()) :: :ok


      


Transfer ownership to a new process.
The current owner will receive {ps, :connected} and the new owner
will start receiving data messages.

          @spec open(binary(), [binary()], keyword()) :: GenServer.on_start()


      


Open a new OS process and return a handle to it.
Returns {:ok, pid} on success or {:error, reason} on failure.
The calling process becomes the owner and will receive messages.
Options
	:cd - the directory to run the command in
	:env - an enumerable of tuples containing environment key-value as binary
	:arg0 - sets the command arg0
	:cgroup_controllers - run the command under the specified cgroup controllers
	:cgroup_base - create a temporary path under the specified cgroup path
	:cgroup_path - explicitly specify a path to use
	:cgroup_sets - set a cgroup controller parameter before running the command
	:delay_to_sigkill - milliseconds before SIGKILL if SIGTERM doesn't work (default 500ms)
	:uid - run the command using the specified uid or username
	:gid - run the command using the specified gid or group
	:stdio_window - flow control window size in bytes (default 10KB)

Example
{:ok, ps} = ProtonStream.open("echo", ["hello"])
receive do
  {^ps, {:data, "hello\n"}} -> :ok
end

          @spec os_pid(pid()) :: non_neg_integer() | nil


      


Return the OS process ID of the child process.

          @spec start_link(module(), binary(), [binary()], term(), keyword()) ::
  GenServer.on_start()

Validate and normalize the options passed to ProtonStream.open/3.
This module is generally not called directly, but it's likely
the source of exceptions if any options aren't quite right. Call validate/4 directly to
debug or check options without invoking a command.

        Validate options and normalize them for invoking commands.

          @type t() :: map()


      


The following fields are always present:
	:command - the command to run
	:args - a list of arguments to the command

The next fields are optional:
	:cd
	:arg0
	:stderr_to_stdout
	:capture_stderr_only
	:parallelism
	:env
	:stdio_window
	:cgroup_controllers
	:cgroup_path
	:cgroup_base
	:delay_to_sigkill
	:cgroup_sets
	:uid
	:gid

          @spec validate(:stream, binary(), [binary()], keyword()) :: t()


      


Validate options and normalize them for invoking commands.

